# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgname=python-aaf2
pkgver=1.7.1
pkgrel=1
pkgdesc='Read and write Advanced Authoring Format (AAF) files'
url='http://pyaaf.readthedocs.io/'
license=(MIT)
arch=(any)
depends=(
        python
        )
makedepends=(
            python-build
            python-installer
            python-setuptools
            python-wheel
            )
source=(https://pypi.python.org/packages/source/p/pyaaf2/pyaaf2-$pkgver.tar.gz)
sha256sums=('e18e5a84bca4ea18c1b9e8384958e2f6d296195c901a448f71f830b22b89c225')

build() {
  cd pyaaf2-$pkgver
  python -m build --wheel --no-isolation
}

package() {
  cd pyaaf2-$pkgver
  python -m installer --destdir="$pkgdir" dist/*.whl
  install -Dm644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname
}
